

OpenDSS PVSystem Element Model

Version 1

Wednesday, February 23, 2011

Figure 1 shows a schematic diagram of the *PVSystem* device model recently implemented into OpenDSS version 7.4.1 at Build 28. This model combines a model of the PV array and the PV inverter into one convenient model to use for distribution system impacts studies.




Figure 1. Block Diagram of the PVSystem Element Model

The present version of the model is useful for simulations with greater than 1s time steps. The model assumed the inverter is able to find the max power point (*mpp*) of the panel quickly. This simplifies the modeling of the individual components (PV panel and inverter) and should be adequate for most interconnection impact studies.

The interface to the circuit model is the same as any Power Conversion (PC) element in the program. It basically appears the same to the circuit model as a Generator or Load or Storage device would, producing or consuming power according to some function. In this case, the active power, P , is a function of the Irradiance, temperature (T), and rated power

at the mpp, P_{mpp} at a selected temperature and an irradiance of 1.0 kW/m^2 . In addition, the efficiency of the inverter at the operating power and voltage is applied.

Reactive Power

Reactive power is specified separately from the active power and may be specified as either fixed kvar values or a fixed power factor value. If the *PF* property is specified, the model holds a constant output power factor until the *PF* property is changed (default mode). If the *kvar* property is specified, the inverter is assumed to attempt to hold that value despite the present value of the panel power. The actual kvar output is dropped if the rated *kVA* of the inverter is exceeded.

These are the only two reactive power modes currently implemented in the present version. Many of the larger PV system inverters can also adjust vars to regulate voltage. Therefore, future versions may have voltage regulation modeling capability. (That capability is nominally in the generic Generator object - model=3).

Loadshapes and Tshapes

For Daily, Yearly, and Duty-cycle simulations, the *Irradiance* and temperature, *T*, base values can be modified by Loadshape and Tshape objects, respectively, corresponding to the type of simulation being performed. This is what enables the model to provide a varying power input for the sequential time simulations.

Both the Yearly and Duty-cycle shapes default to the shape defined for the Daily shape if they are not specified. If the Daily shape is not defined, it defaults to a constant multiplier of 1.0.

XYCurve Objects for Power-Temperature and Efficiency

The PVsystem model uses *XYcurve* objects to describe certain characteristics of the PV panels and inverters. XYcurve objects are new with this version of OpenDSS. You may enter x-y curves as either an array of points or as separate arrays of x and y values. The following two examples are equivalent:

```
// curve in separate x, y array  
New XYCurve.MyEff npts=4 xarray=[0.1 0.2 0.4 1.0] yarray=[0.86 0.9 0.93 0.97]
```

```
// curve as array of x,y values, in sequence  
New XYCurve.MyEff npts=4 points=[0.1, 0.86 0.2, 0.9 0.4, 0.93 1.0, 0.97]
```

For the Points property, the values in the array may be separated by either commas or white space (space or tab). Commas are used in the above example to emphasize the x,y points.

XYcurve objects are interpolated linearly between defined points to determine the actual value. For curves used in the PVsystem model, it is usually sufficient to enter only 4 or 5 points because the curves are relatively smooth and monotonic.

An array of points is entered to describe how the P_{mpp} varies with *T* relative to the temperature chosen for the rated P_{mpp} at 1 kW/m^2 . This is a per unit factor for discounting the panel power output for temperature. The factor is 1.0 for the temperature for which

the P_{mpp} is defined. Then it normally declines for higher temperatures and increases for lower temperatures.

An array of points is also used to represent the efficiency curve for the inverter. While this is a family of curves depending on the dc bus voltage, the model uses only a single curve at this time, using a curve near the typical operating voltage of a given array. This model may be made more sophisticated in future revisions, but this simplified model appears adequate for distribution impact studies.

State Variables

Like other PC elements in the program, the PVsystem element has internal state variables that can be queried and observed. The present variable names are:

1. **'Irradiance'** This is the net irradiance after applying the load shape factor for the present simulation mode.
2. **'PanelkW'** This is the net power, kW, coming out of the panel taking into consideration the irradiance and the temperature.
3. **'P_TFactor'** The factor interpolated from the Power-Temperature curve for the present solution. This is applied to the base P_{mpp} at the reference temperature to compute the panel kW.
4. **'Efficiency'** The efficiency factor for the inverter.

You can observe the values of these state variables during simulations by placing a Monitor element, mode 3, on the terminal of the PVsystem element. Also, for a static load condition, you can issue the “Show Variables” command to see the values of all state variables in the system.

Using the Model

The basic data for the model are:

- An average P_{mpp} for the panel at 1 kW/m^2 irradiance at a constant panel temperature such as 25C or 50C.
- The per unit variation of P_{mpp} vs Temperature at 1 kW/m^2 irradiance.
- A representative efficiency curve for the inverter, per unit efficiency vs per unit power.

Figure 2 shows the general form of i-v curves for a PV panel at a constant temperature. The max power point, mpp, is shown on each curve. The ratio of P_{mpp} to irradiance at a constant temperature varies a few percent over a practical range, but is often close enough to being constant for the purpose of distribution impacts studies.




Figure 2. Typical form of i-v curves for different values of irradiance at a constant temperature, with max power points indicated.

The model parameters are specified for a relatively high irradiance value of 1000 W/m² so that the model is expected to be more accurate at higher power output where issues like voltage rise would be most important.

Given an irradiance value, the panel output is then discounted by a factor depending on the panel temperature. For example, if the P_{mpp} is supplied for a panel temperature of 25°C, the power vs temperature curve might be similar to that shown in Figure 3. As the panel temperature increases from 25 deg to 75°C, the power drops off by about 22%. Thus the panel output max power output is estimated by

$$\text{Panel kW} = P_{mpp} \text{ (in kW @ } 1\text{ kW/m}^2 \text{ and } 25\text{ }^\circ\text{C)} * \text{Irradiance (in kW/m}^2) * \text{Factor (@actual T)}$$




Figure 3. Example Power - Temperature variation for 1 kW/m² irradiance




Figure 4. Efficiency vs Inverter Power

Finally, the model then assumes the inverter is able to find the mpp within the simulation time step. The inverter output power is determined by applying the efficiency from a curve like Figure 4. This is a typical efficiency curve from the expected dc operating voltage at high irradiance values. Eventually, if it is determined to be important, the model might be adapted to include a family of efficiency curves.

Static (Snapshot) Solutions

- Set the value of Irradiance in kW/m^2 and the panel temperature.
- Set the PF or kvar if the default of unity power factor is not satisfactory
- Solve

The program will iterate to a solution that achieves the requisite output power. If necessary, the kvar output will be cut back to get under the kVA rating limit.

Sequential Time Simulations

- Create and assign a Loadshape object to either the Duty, Daily, or Yearly property as appropriate. This loadshape is multiplied by the base value specified for the Irradiance.
- Create and assign a Tshape object to either the TDuty, TDaily, or TYearly property as appropriate. This describes the panel temperature that corresponds to the irradiance Loadshape.
- Set the solution mode, time step size, and number of steps.
- Solve

- Display monitor results

Example Script for Exercising the PVSystem Model

This example defines a PV system with a panel P_{mp} of 500 kW at 1 kW/m² irradiance and a panel temperature of 25°C. The inverter is rated at 500 kVA. A PF of 1.0 is assumed for this example.

```
clear

New Circuit.PVSystem basekv=12.47 Isc3=1000 Isc1=900

// P-T curve is per unit of rated Pmp vs temperature
// This one is for a Pmp stated at 25 deg
New XYCurve.MyPvsT npts=4 xarray=[0 25 75 100] yarray=[1.2 1.0 0.8 0.6]

// efficiency curve is per unit eff vs per unit power
New XYCurve.MyEff npts=4 xarray=[.1 .2 .4 1.0] yarray=[.86 .9 .93 .97]

// per unit irradiance curve (per unit if "irradiance" property)
New Loadshape.MyIrrad npts=24 interval=1 mult=[0 0 0 0 0 0 .1 .2 .3 .5 .8 .9
1.0 1.0 .99 .9 .7 .4 .1 0 0 0 0 0 0]

// 24-hr temp shape curve
New Tshape.MyTemp npts=24 interval=1 temp=[25, 25, 25, 25, 25, 25, 25, 25, 35,
40, 45, 50 60 60 55 40 35 30 25 25 25 25 25 25]

// **** plot tshape object=mytemp

// take the default line
New Line.line1 Bus1=sourcebus bus2=PVbus Length=2

// pv definition
New PVSystem.PV phases=3 bus1=PVbus kV=12.47 kVA=500 irrads=0.8 Pmp=500
~ temperature=25 PF=1 effcurve=Myeff P-TCurve=MyPvsT
~ Daily=MyIrrad TDaily=MyTemp

set voltagebases=[12.47]
calc

solve ! solves at the specified irradiance and temperature

new monitor.m1 PVSystem.PV 1 mode=1 ppolar=no
new monitor.m2 PVSystem.PV 1

solve
solve mode=daily

show mon m1
show mon m2

Export monitors m1
Plot monitor object= m1 channels=(1 )
Export monitors m2
Plot monitor object= m2 channels=(1 ) base=[7200]
Export monitors m2
Plot monitor object= m2 channels=(9 )
```

Properties

Property	Description
(1) phases	Number of Phases, this PVSystem element. Power is evenly divided among phases.
(2) bus1	Bus to which the PVSystem element is connected. May include specific node specification.
(3) kv	Nominal rated (1.0 per unit) voltage, kV, for PVSystem element. For 2- and 3-phase PVSystem elements, specify phase-phase kV. Otherwise, specify actual kV across each branch of the PVSystem element. If 1-phase wye (star or LN), specify phase-neutral kV. If 1-phase delta or phase-phase connected, specify phase-phase kV.
(4) irradiance	Get/set the present irradiance value in kW/sq-m. Used as base value for shape multipliers. Generally entered as peak value for the time period of interest and the yearly, daily, and duty load shape objects are defined as per unit multipliers (just like Loads/Generators).
(5) Pmpp	Get/set the rated max power of the PV array for 1.0 kW/sq-m irradiance and a user-selected array temperature. The P-TCurve should be defined relative to the selected array temperature.
(6) Temperature	Get/set the present Temperature. Used as fixed value corresponding to PTCurve property. A multiplier is obtained from the Pmpp-Temp curve and applied to the nominal Pmpp from the irradiance to determine the net array output.
(7) pf	Nominally, the power factor for the output power. Default is 1.0. Setting this property will cause the inverter to operate in CONSTANT POWER FACTOR MODE. Enter negative when kW and kvar have opposite signs.
	A positive power factor signifies that the PVSystem element produces vars as is typical for a generator.
(8) conn	= {wye LN delta LL}. Default is wye.
(9) kvar	Get/set the present kvar value. Setting this property forces the inverter to operate in CONSTANT KVAR MODE.
(10) kVA	kVA rating of inverter. Used as the base for Dynamics mode and Harmonics mode values.
(11) %Cutin	% cut in power -- % of kVA rating of inverter. When the inverter is OFF, the power from the array must be greater than this for the inverter to turn on.
(12) %Cutout	% cut out power -- % of kVA rating of inverter. When the inverter is ON, the inverter turns OFF when the power from the array drops below this value.

(13) EffCurve	An XYCurve object, previously defined, that describes the PER UNIT efficiency vs PER UNIT of rated kVA for the inverter. Inverter output power is discounted by the multiplier obtained from this curve.
(14) P-TCurve	An XYCurve object, previously defined, that describes the PV array PER UNIT Pmpp vs Temperature curve. Temperature units must agree with the Temperature property and the Temperature shapes used for simulations. The Pmpp values are specified in per unit of the Pmpp value for 1 kW/sq-m irradiance. The value for the temperature at which Pmpp is defined should be 1.0. The net array power is determined by the irradiance * Pmpp * f(Temperature)
(15) %R	Equivalent percent internal resistance, ohms. Default is 0. Placed in series with internal voltage source for harmonics and dynamics modes. Use a combination of %IdlekW and %EffCharge and %EffDischarge to account for losses in power flow modes.
(16) %X	Equivalent percent internal reactance, ohms. Default is 50%. Placed in series with internal voltage source for harmonics and dynamics modes. (Limits fault current to 2 pu.) Use %Idlekvar and kvar properties to account for any reactive power during power flow solutions.
(17) model	Integer code (default=1) for the model to use for power output variation with voltage. Valid values are: 1: PVSystem element injects a CONSTANT kW, kvar at specified power factor or kvar value 2: PVSystem element is modeled as a CONSTANT ADMITTANCE. 3: Compute load injection from User-written Model.
(18) Vminpu	Default = 0.90. Minimum per unit voltage for which the Model is assumed to apply. Below this value, the load model reverts to a constant impedance model.
(19) Vmaxpu	Default = 1.10. Maximum per unit voltage for which the Model is assumed to apply. Above this value, the load model reverts to a constant impedance model.
(20) yearly	Dispatch shape to use for YEARLY simulations. Must be previously defined as a Loadshape object. If this is not specified, the Daily dispatch shape, If any, is repeated during Yearly solution modes. In the default dispatch mode, the PVSystem element uses this loadshape to trigger State changes.
(21) daily	Dispatch shape to use for DAILY simulations. Must be previously defined as a Loadshape object of 24 hrs, typically. In the default dispatch mode, the PVSystem element uses this loadshape to trigger State changes.
(22) duty	Load shape to use for DUTY cycle dispatch simulations such as for solar ramp rate studies. Must be previously defined as a Loadshape object. Typically would have time intervals of 1-5 seconds. Designate the number of points to solve using the Set Number=xxxx command. If there are fewer points in the actual shape, the shape is assumed to repeat.
(23) Tyearly	Temperature shape to use for YEARLY simulations. Must be previously defined as a TShape object. If this is not specified, the Daily dispatch shape, If any, is repeated during Yearly solution modes. The PVSystem element uses this TShape to determine the Pmpp from the Pmpp vs T curve. Units must agree with the

	Pmpp vs T curve.
(24) Tdaily	Temperature shape to use for DAILY simulations. Must be previously defined as a TShape object of 24 hrs, typically. The PVSystem element uses this TShape to determine the Pmpp from the Pmpp vs T curve. Units must agree with the Pmpp vs T curve.
(25) Tduty	Temperature shape to use for DUTY cycle dispatch simulations such as for solar ramp rate studies. Must be previously defined as a TShape object. Typically would have time intervals of 1-5 seconds. Designate the number of points to solve using the Set Number=xxxx command. If there are fewer points in the actual shape, the shape is assumed to repeat. The PVSystem model uses this TShape to determine the Pmpp from the Pmpp vs T curve. Units must agree with the Pmpp vs T curve.
(26) class	An arbitrary integer number representing the class of PVSystem element so that PVSystem values may be segregated by class.
(27) UserModel	Name of DLL containing user-written model, which computes the terminal currents for Dynamics studies, overriding the default model. Set to "none" to negate previous setting.
(28) UserData	String (in quotes or parentheses) that gets passed to user-written model for defining the data required for that model.
(29) debugtrace	{ Yes No } Default is no. Turn this on to capture the progress of the PVSystem model for each iteration. Creates a separate file for each PVSystem element named "PVSystem_name.CSV".
(30) spectrum	Name of harmonic voltage or current spectrum for this PVSystem element. Current injection is assumed for inverter. Default value is "default", which is defined when the DSS starts.